Come definire una funzione ...

In[1]:=

```
f[x_] := E^(-x^2)          (* 1 argomento *)
```

In[2]:=

```
? f
Global`f
f[x_] := E^(-x^2)
```

In[3]:=

```
Plot[f[x],{x,-3,3}]
```



Out[3]=

```
-Graphics-
```

In[4]:=

```
f[x]
```

Out[4]=

$$E^{-x^2}$$

In[5]:=
```
f'[x]
```
Out[5]=
$$-2\,x\,e^{-x^2}$$
In[6]:=
```
f''[x]
```
Out[6]=
$$-2\,e^{-x^2}+4\,x^2\,e^{-x^2}$$
In[7]:=
```
Integrate[f[x],x]
```
Out[7]=
$$\frac{\sqrt{\pi}\,\mathrm{Erf}[x]}{2}$$
In[8]:=
```
Integrate[f[x],{x,-3,3}]
```
Out[8]=
$$\sqrt{\pi}\,\mathrm{Erf}[3]$$
In[9]:=
```
N[%]
```
Out[9]=
$$1.77241$$

In[10]:=
```
Integrate[f[x],{x,-Infinity,Infinity}]
```
Out[10]=
```
Sqrt[Pi]
```
In[11]:=
```
g[x_] := f[x]/Sqrt[Pi]
```
In[12]:=
```
Integrate[g[x],{x,-Infinity,Infinity}]
```
Out[12]=
```
1
```
In[13]:=
```
g[x_,y_] := g[x] g[y]   (* 2 argomenti *)
```
In[14]:=
```
Plot3D[g[x,y],{x,-3,3},{y,-3,3},
  PlotRange -> All,PlotPoints -> {30,30}]
```



Out[14]=
```
-SurfaceGraphics-
```

In[15]:=
```
? g
Global`g
g[x_] := f[x]/Sqrt[Pi]

g[x_, y_] := g[x]*g[y]
```
In[16]:=
```
Derivative[1,0][g][x,y]
```
Out[16]=
$$\frac{-2\,E^{-x^2-y^2}\,x}{Pi}$$
In[17]:=
```
Derivative[1,1][g][x,y]
```
Out[17]=
$$\frac{4\,E^{-x^2-y^2}\,x\,y}{Pi}$$
In[18]:=
```
Integrate[g[x,y],
  {x,-Infinity,Infinity},
  {y,-Infinity,Infinity}]
```
Out[18]=
```
1
```
In[19]:=
```
g[x__] := Apply[Times,Map[g,{x}]]
                              (* n argomenti *)
```

In[21]:=

```
g[x,y,z]
```

Out[21]=

$$\frac{E^{-x^2-y^2-z^2}}{\text{Pi}^{3/2}}$$

In[22]:=

```
Integrate[%,{x,-Infinity,Infinity},
          {y,-Infinity,Infinity},
          {z,-Infinity,Infinity}]
```

Out[22]=

```
1
```

... vettoriale e/o con parametri

In[23]:=
```
Norma[v_] := Sqrt[v . v]   (* R^n -> R *)
```
```
General::spell1:
   Possible spelling error: new symbol
     name "Norma"
     is similar to existing symbol
    "Normal".
```
In[24]:=
```
Norma[x]
```
Out[24]=
$$\text{Sqrt}[x \cdot x]$$
In[25]:=
```
Norma[{x}]
```
Out[25]=
$$\text{Sqrt}[x^2]$$
In[26]:=
```
Norma[{x,y}]
```
Out[26]=
$$\text{Sqrt}[x^2 + y^2]$$
In[27]:=
```
Norma[{x,y,z}]
```
Out[27]=
$$\text{Sqrt}[x^2 + y^2 + z^2]$$
In[28]:=
```
Distanza[p_,q_] := Norma[p - q]
```

In[29]:=

```
Distanza[{x1,y1},{x2,y2}]
```

Out[29]=

$$\text{Sqrt}[(x1 - x2)^2 + (y1 - y2)^2]$$

In[30]:=

```
Circ[r_][t_] := r {Cos[t],Sin[t]}
                                    (* R -> R^2 *)
```

In[32]:=

```
ParametricPlot[Circ[1][t]//Evaluate,
  {t,0,2Pi},AspectRatio -> Automatic]
```



Out[32]=

```
-Graphics-
```

In[33]:=

```
Circ[2]
```

Out[33]=

$$\text{Circ}[2]$$

In[34]:=

```
Circ[2][t]
```

Out[34]=

```
{2 Cos[t], 2 Sin[t]}
```

In[35]:=

```
Norma[Circ[r][t]] // Simplify
```

Out[35]=

$$\sqrt{r^2}$$

In[36]:=

```
PowerExpand[%]
```

Out[36]=

```
r
```

In[37]:=

```
Ellisse[a_,b_][t_] := {a,b} Circ[1][t]
```

In[38]:=

```
ParametricPlot[
  Ellisse[2,1][t]//Evaluate,
  {t,0,2Pi},AspectRatio -> Automatic]
```



Out[38]=

```
-Graphics-
```

In[39]:=

```
T[v_][x_] := x + v        (* R^n -> R^n *)
```

In[40]:=

```
Circ[r_,c_][t_] := T[c][Circ[r][t]]
```

In[41]:=

```
Ellisse[a_,b_,c:{_,_}][t_] :=
  T[c][Ellisse[a,b][t]]
```

In[42]:=

```
ParametricPlot[
  Table[Ellisse[2,1,{k,-k}][t],
        {k,0,1,.1}] // Evaluate,
  {t,0,2Pi},AspectRatio -> Automatic]
```



Out[42]=

-Graphics-

In[43]:=

```
M[a_] := {{Cos[a],-Sin[a]},
          {Sin[a],Cos[a]}}
```

In[44]:=

```
MatrixForm[M[a]]
```

Out[44]//MatrixForm=

```
Cos[a]    -Sin[a]

Sin[a]    Cos[a]
```

In[45]:=

```
R[a_][{x_,y_}] = M[a] . {x,y}
```

Out[45]=

```
{x Cos[a] - y Sin[a],

  y Cos[a] + x Sin[a]}
```

In[46]:=

```
Ellisse[a_,b_,c_:{0,0},ang_][t_] :=
  T[c][R[ang][Ellisse[a,b][t]]]
```

In[47]:=

```
? Ellisse
Global`Ellisse
Ellisse[a_, b_][t_] := {a, b}*Circ[1][t]

Ellisse[a_, b_, c:{_, _}][t_] :=
  T[c][Ellisse[a, b][t]]

Ellisse[a_, b_, c_:{0, 0}, ang_][t_] :=
  T[c][R[ang][Ellisse[a, b][t]]]
```

In[48]:=

```
ParametricPlot[
  {Ellisse[2,1][t],
   Ellisse[2,1,Pi/4][t],
   Ellisse[2,1,{4,2}][t],
   Ellisse[2,1,{4,2},Pi/4][t]}//Evaluate,
  {t,0,2Pi},AspectRatio -> Automatic]
```



Out[48]=

-Graphics-

Rappresentazioni grafiche

In[49]:=

```
Plot3D[Sin[x] Cos[y],
  {x,-Pi,Pi},{y,-Pi,Pi},
  PlotPoints -> {60,60}]
```



Out[49]=

-SurfaceGraphics-

In[50]:=

```
Show[%,Mesh -> False,Boxed -> False]
```



Out[50]=

-SurfaceGraphics-

In[51]:=

```
Show[%,Axes -> False,
      ColorFunction -> Hue]
```



Out[51]=

```
-SurfaceGraphics-
```

In[52]:=

```
DensityPlot[Sin[x] Cos[y],
  {x,-Pi,Pi},{y,-Pi,Pi},
  PlotPoints -> {60,60},Mesh -> False]
```



Out[52]=

```
-DensityGraphics-
```

In[53]:=

```
ContourPlot[Sin[x] Cos[y],
  {x,-Pi,Pi},{y,-Pi,Pi},
  PlotPoints -> {40,40},Contours -> 20]
```



Out[53]=

-ContourGraphics-

In[54]:=

```
ParametricPlot3D[
  {3 Cos[2Pi t],3 Sin[2Pi t],t},
  {t,-2,2},PlotPoints -> 300]
```



Out[54]=

-Graphics3D-

In[55]:=

```
ParametricPlot3D[
  {Cos[t](3 + Cos[s]),
   Sin[t](3 + Cos[s]),Sin[s]},
  {t,0,2Pi},{s,0,2Pi},
  PlotPoints -> {60,30}]
```



Out[55]=

```
-Graphics3D-
```

In[56]:=

```
ParametricPlot3D[
  {Cos[t](3 + Cos[s]),
   Sin[t](3 + Cos[s]),Sin[s],EdgeForm[]},
  {t,0,2Pi},{s,0,2Pi},
  PlotPoints -> {60,30},
  Boxed->False,Axes->False]
```



Out[56]=

-Graphics3D-

In[57]:=

```
ParametricPlot3D[
  {Cos[t](3 + Cos[s]),
   Sin[t](3 + Cos[s]),Sin[s],
   Hue[(t+s)/Pi]},
  {t,0,2Pi},{s,0,2Pi},
  PlotPoints -> {60,30},
  Boxed->False,Axes->False,
  Lighting->False]
```



Out[57]=

-Graphics3D-

Funzioni derivata e primitiva

In[58]:=
```
Derivata[f_][x_] :=
  Module[{h},
    Limit[(f[x+h] - f[x])/h,h -> 0]]
```
In[59]:=
```
Derivata[Sin][x]
```
Out[59]=
```
Cos[x]
```
In[60]:=
```
Derivata[Sin]
```
Out[60]=
```
Derivata[Sin]
```
In[61]:=
```
Derivata1[f_] :=
  Function[x,Derivata[f][x]]
```
In[62]:=
```
Derivata2[f_] :=
  Evaluate[Derivata[f][#]]&
```
In[63]:=
```
{Derivata1[Sin][x],Derivata2[Sin][x]}
```
Out[63]=
```
{Cos[x], Cos[x]}
```
In[64]:=
```
{Derivata1[Sin],Derivata2[Sin],Sin'}
```
Out[64]=
```
{Function[x$, Derivata[Sin][x$]],
  Cos[#1] & , Cos[#1] & }
```

In[65]:=

```
Primitiva[f_,x0_:0][x_] :=
  Module[{h},Integrate[f[h],{h,x0,x}]]
```

In[66]:=

```
Primitiva[Sin][x]
```

Out[66]=

```
1 - Cos[x]
```

In[67]:=

```
Primitiva1[f_,x0_:0] :=
  Function[x,Primitiva[f,x0][x]]
```

In[68]:=

```
Primitiva2[f_,x0_:0] :=
  Evaluate[Primitiva[f,x0][#]]&
```

In[69]:=

```
{Primitiva1[Sin],Primitiva2[Sin]}
```

Out[69]=

```
{Function[x$, Primitiva[Sin, 0][x$]],
  1 - Cos[#1] & }
```

Funzioni "pure"

In[70]:=
```
? g
Global`g
g[x_] := f[x]/Sqrt[Pi]

g[x_, y_] := g[x]*g[y]

g[x__] := Times @@ g /@ {x}
```
In[71]:=
```
h = Function[x,f[x]/Sqrt[Pi]]
```
Out[71]=
$$\text{Function}\left[x, \frac{f[x]}{\text{Sqrt}[\text{Pi}]}\right]$$
In[72]:=
```
k = f[#]/Sqrt[Pi]&
```
Out[72]=
$$\frac{f[\#1]}{\text{Sqrt}[\text{Pi}]} \&$$
In[73]:=
```
{g,h,k}
```
Out[73]=
$$\left\{g, \text{Function}\left[x, \frac{f[x]}{\text{Sqrt}[\text{Pi}]}\right], \frac{f[\#1]}{\text{Sqrt}[\text{Pi}]} \&\right\}$$

In[74]:=
```
{h[x],k[x]}
```
Out[74]=
$$\left\{\frac{1}{E^{x^2}\ \text{Sqrt}[\text{Pi}]},\ \frac{1}{E^{x^2}\ \text{Sqrt}[\text{Pi}]}\right\}$$
In[75]:=
```
h2 = Function[{x,y},Evaluate[h[x] h[y]]]
```
Out[75]=
$$\text{Function}\left[\{x,\ y\},\ \frac{E^{-x^2-y^2}}{\text{Pi}}\right]$$
In[76]:=
```
k2 = Evaluate[h[#1] h[#2]]&
```
Out[76]=
$$\frac{E^{-\#1^2-\#2^2}}{\text{Pi}}\ \&$$
In[77]:=
```
kn = Apply[Times,Map[k,{##}]]&
```
Out[77]=
```
Times @@ k /@ {##1} &
```
In[78]:=
```
km = Evaluate[Apply[Times,Map[k,{##}]]]&
```
Out[78]=
$$\frac{1}{E^{\#\#1^2}\ \text{Sqrt}[\text{Pi}]}\ \&$$

In[79]:=
```
{k2[x,y],kn[x,y],km[x,y]}
```
Out[79]=
```
      2    2      2    2
    -x  - y    -x  - y
  E           E                1
 {----------, ----------, --------------}
     Pi           Pi            2
                                y
                               x
                              E    Sqrt[Pi]
```
In[80]:=
```
? Function
```
```
Function[body] or body& is a pure
   function. The formal parameters are #
   (or #1), #2, etc. Function[x, body]
   is a pure function with a single
   formal parameter x. Function[{x1, x2,
   ... }, body] is a pure function with
   a list of formal parameters.
```
In[81]:=
```
Circ1[r_] :=
  Function[t,Evaluate[Circ[r][t]]]
```
In[82]:=
```
Circ2 := Function[r,Evaluate[Circ1[r]]]
```
In[83]:=
```
{Circ,Circ1,Circ2}
```
Out[83]=
```
{Circ, Circ1, Function[r,
   Function[t$, {r Cos[t$], r Sin[t$]}]]
   }
```

In[84]:=
```
{Circ[r],Circ1[r],Circ2[r]}
```
Out[84]=
```
{Circ[r], Function[t$,
    {r Cos[t$], r Sin[t$]}],
   Function[t$, {r Cos[t$], r Sin[t$]}]}
```
In[85]:=
```
{Circ[r][t],Circ1[r][t],Circ2[r][t]}
```
Out[85]=
```
{{r Cos[t], r Sin[t]},
   {r Cos[t], r Sin[t]},
   {r Cos[t], r Sin[t]}}
```
In[86]:=
```
Table[Random[Real,{-1,1}],{8000},{2}];
```
In[87]:=
```
pi = Select[%,(Norma[#] <= 1)&];
```
In[88]:=
```
pe = Select[%%,(Norma[#] > 1)&];
```
In[89]:=
```
N[Length[pi]/2000]
```
Out[89]=
```
3.1485
```
In[90]:=
```
ListPlot[pi // Evaluate,
          PlotStyle->Hue[.7],
          DisplayFunction->(#&)]
```
Out[90]=
```
-Graphics-
```

In[91]:=

```
ListPlot[pe // Evaluate,
         PlotStyle->Hue[1],
         DisplayFunction->(#&)]
```

Out[91]=

-Graphics-

In[92]:=

```
Show[{%,%%},DisplayFunction->
            $DisplayFunction,
            AspectRatio->Automatic,
            Frame->True]
```



Out[92]=

-Graphics-

Oggetti grafici

In[93]:=
```
P := Polygon[{{0,0},{1,0},{1,1},{0,1}}]
```
In[94]:=
```
Q := Line[{{0,0},{1,0},
          {1,1},{0,1},{0,0}}]
```
In[95]:=
```
Graphics[{P,Q /. {x_,y_} -> {x+2,y}}]
```
Out[95]=

-Graphics-

In[96]:=
```
Show[%,AspectRatio->Automatic]
```



Out[96]=

-Graphics-

In[97]:=
```
InputForm[%]
```
Out[97]//InputForm=
```
Graphics[{Polygon[{{0, 0}, {1, 0},
     {1, 1}, {0, 1}}],
   Line[{{2, 0}, {3, 0}, {3, 1},
     {2, 1}, {2, 0}}]},
  {AspectRatio -> Automatic}]
```

In[98]:=

```
Graphics3D[{P /. {x_,y_} -> {x,y,0},
            Q /. {x_,y_} -> {x+2,y,0}}]
```

Out[98]=

-Graphics3D-

In[99]:=

```
Show[%,PlotRange -> {All,All,{-.5,.5}}]
```



Out[99]=

-Graphics3D-

In[100]:=

```
Graphics3D[
  Table[P /. {x_,y_} ->
              {x+h,y+k,(x h + y k)/4},
        {h,-8,8},{k,-8,8}]]
```

Out[100]=

-Graphics3D-

In[101]:=

```
Show[%,Boxed -> False]
```



Out[101]=

-Graphics3D-

In[102]:=

```
Graphics[
  Table[{{PointSize[.002 x],
          Point[{x,1}]},
         {Thickness[.002 x],
          Line[{{x,2},{x,3}}]}},{x,10}],
  Frame->True,FrameTicks->{Range[10],{}},
  PlotRange->{All,{.5,3.5}}]
```

Out[102]=

-Graphics-

In[103]:=

```
Show[%,AspectRatio->Automatic]
```



Out[103]=

```
-Graphics-
```

In[104]:=

```
Graphics[
  Table[{Hue[h],Line[{{h,0},{h,.2}}]},
        {h,0,1,.002}],
  Frame->True,FrameTicks->{Automatic,{}}]
```

Out[104]=

```
-Graphics-
```

In[105]:=

```
Show[%,AspectRatio->Automatic]
```



Out[105]=

```
-Graphics-
```